

SE [SebastianOrre](#)

Created on May 31, 2017

## Question Info

# Creating a blank sources list, and managing source lists

Last updated June 20, 2019

Hi!

Views 1,011

I'm working with several projects simultaneously in school. All of them requires reports with several references and sources. And since the reports are in different subjects, the sources are not overlapping in any way. If I wouldn't have deleted my old sources manually I would have close to 50 sources in my masters list by now, which is a real pain when trying to find the ones to copy to "current list".

Applies to:



Word /

[Microsoft Office Programming](#)  
/ [Office 2010](#)

Since my list of sources (.xml-file) for the different subjects doesn't overlap, I don't want to see all my sources all the time. Basically I want a separate source list for every subject/report.

I quickly figured out that I can manually make copies of the .xml in the bibliography folder, but I can't manage my .xml-files at all within Word (MS Office Prof Plus 2010, on this computer). And I haven't understood yet how transferals of the reports between computers (or syncing with OneDrive). Like, if I mail my report.docx to a friend, can I make Word understand what source list to bring along and what to leave behind?

Might it be like this: The "current list" is "built in" to the .docx-file? So when I email my report to a friend, only the sources in the "current list" comes along?

Another closely related issue to this is that I have an option as adding only "cited sources" when adding my reference list to my report, but it always adds all my sources (from the field to the right - current list) to the report, so I have to delete them manually from my current list. Which is also a pain because I might need them later...

Wow, I really need to practice shortening down my questions...

Thank you!  
Sebastian

🔒 This thread is locked. You can follow the question or vote as helpful, but you cannot reply to this thread.

[I have the same question \(5\)](#)    · · ·

## Replies (1)

### A. User

Replied on June 1, 2017 

Just so you know, I don't work much with Bibliographies, but the following notes may help you anyway.

<<

but I can't manage my .xml-files at all within Word (MS Office Prof Plus 2010, on this computer

>>

Well, "management" is limited to being able to specify the current "master list". You can do that in the UI in the "Source Manager" by clicking on the Browse button and locating and selecting the file you want.

You can do it programmatically in VBA by using Application.LoadMasterList and specifying the full name (path+name) of the relevant .xml file.

(If you are collaborating with people using Mac desktop versions of Word, please be aware that \*they\* cannot do either of those things).

But in principle that might allow you to code something to make list switching a bit simpler. For example, you might have a different template for the documents that are associated with each of your master lists, and put VBA code in the template that would select the correct master list when you create or open a document based on that template. Or you could store the full path and file name of the relevant master source list in each document (perhaps in a custom document property or a document variable,

and use a bit of code to load the correct master list.

<<

Might it be like this: The "current list" is "built in" to the .docx-file? So when I email my report to a friend, only the sources in the "current list" comes along?

>>

That is exactly how it works. If you make a copy of your .docx, rename it to .zip, open the .zip and navigate through the files, you should see a folder called customXml. In there, there will be one or more files called item1.xml, item2.xml, etc. One of these will contain your "current list". So the current lists are in effect independent of the master list, except for the fact that each entry in a master list has to have a unique tag, and may be assigned a unique GUID (such as {D952E393-FBD4-3E40-8753-6A0AE119A15A} ), which Word may use to work out what is and isn't used.

<<

Another closely related issue to this is that I have an option as adding only "cited sources" when adding my reference list to my report, but it always adds all my sources (from the field to the right - current list) to the report, so I have to delete them manually from my current list

>>

That is also correct. Unfortunately, the Works Cited and Bibliography lists in Word have always included all the works in your current list, whether you have actually cited them or not. I have never understood why, especially as some of the online documentation suggests that they should be different.

So to make the two things different, on the face of it you would have to generate the Bibliography first (i.e. including entries you haven't actually cited), then convert the BIBLIOGRAPHY field results to text so they are "fixed". Then remove the uncited entries from the current list, and generate the Works Cited list.

Programmatically, you could consider using code to copy the entire master list to the current list when you start working on a new document, like this:

```
Sub copyMasterListToCurrentList()  
Dim s As String
```

```
Dim src As Word.Source
Dim srCs As Word.Sources
'Debug.Print Application.Bibliography.Sources.Count,
ActiveDocument.Bibliography.Sources.Count
```

```
For Each src In Application.Bibliography.Sources
    If TagIsNew(src.tag) Then ' may need to check guid
    also.
        ActiveDocument.Bibliography.Sources.Add src.XML
    End If
Next
'Debug.Print Application.Bibliography.Sources.Count,
ActiveDocument.Bibliography.Sources.Count
End Sub
```

```
Function TagIsNew(tag As String) As Boolean
Dim src As Word.Source
For Each src In ActiveDocument.Bibliography.Sources
    If tag = src.tag Then
        TagIsNew = False
        Exit Function
    End If
Next
TagIsNew = True
End Function
```

NB, you may find you need to do a bit more work on these routines - they are just starting points.

When you are ready to produce your bibliography and cited works lists, you would remove all the entries from the current list that you did not want in the Bibliography.

Generate the Bibliography, then convert it to static text.

Then you can remove the uncited sources from the current list using

```
Sub removeUncitedSourcesFromCurrentList()
Dim src As Word.Source
Dim L As Long
With ActiveDocument.Bibliography.Sources
    Debug.Print Application.Bibliography.Sources.Count,
.Count
    For L = .Count To 1 Step -1
        Set src = .Item(L)
        If src.Cited = False Then
```

```

        src.Delete
    End If
    Set src = Nothing
Next
Debug.Print Application.Bibliography.Sources.Count,
.Count
End With
End Sub

```

then generate your Cited Works list.

The problem with this is of course that if you need to make further changes, you will need to re-include the sources that you want in the Bibliography, and currently there is no way to record which ones you want. One approach might be to store the Tag names in Document variables, like this

```

Sub removeUncitedSourcesFromCurrentList()
Dim src As Word.Source
Dim taglist As String
Dim L As Long
taglist = " "
With ActiveDocument.Bibliography.Sources
    Debug.Print Application.Bibliography.Sources.Count,
.Count
    For L = .Count To 1 Step -1
        Set src = .Item(L)
        If src.Cited = False Then
            taglist = taglist & src.tag & " "
            src.Delete
        End If
        Set src = Nothing
    Next
    ActiveDocument.Variables("BIBL_taglist").Value =
taglist
    Debug.Print Application.Bibliography.Sources.Count,
.Count
End With
End Sub

```

then you can restore the uncited sources using something like this:

```

Sub restoreUncitedSourcesToCurrentList()
Dim a() As String
Dim src As Word.Source
Dim taglist As String
Dim L As Long
With ActiveDocument.Bibliography.Sources

```

```

Debug.Print Application.Bibliography.Sources.Count,
.Count
taglist =
ActiveDocument.Variables("BIBL_taglist").Value
If Trim(taglist) <> "" Then
a = Split(taglist, " ")
For L = LBound(a) To UBound(a)
If TagIsNew(a(L)) Then
For Each src In Application.Bibliography.Sources
If a(L) = src.tag Then
.Add src.XML
Exit For
End If
Next
End If
Next
End If
' you might erase BIBL_taglist, but it's doing no harm.
Debug.Print Application.Bibliography.Sources.Count,
.Count
End With
End Sub

```

Another approach (which seems to me to suffer from a number of problems) would be to make use of the \f switch in the Bibliography. That lets you specify a single LCID (e.g. 1031 for German/Germany). So if you set all the LCIDs of the cited works to the same LCID, and set all the uncited works to another LCID, you could use the \f switch in Works cited and not in the Bibliography. I can see a couple of ways that might make that a reasonably workable approach, but it's probably only worth pursuing if you never ever need to use LCIDs for anything else.

Anyway, back to the question of what to do when you share your .docx. In essence, if you want your recipient to have access to all the sources in the appropriate list, you'll either have to copy the entire list to the document, or separately share the .xml file.

If you do the latter, the recipient will need to know about using Citation Source Manager to choose a different master .xml file, in order to be able to copy the entries themselves. And beware if you are working with people who are using the Mac desktop version of Word, because that option does not exist in the Mac version of the Citation Source Manager. In that case, a user would need to know where to find the

sources.xml and now to rename it and replace it with whatever .xml file you share with them.

3 people were helped by this reply ·  
Did this solve your problem? [Yes](#) [No](#)

What's new	Microsoft Store	Education	Enterprise	Developer	Company
Surface Pro 6	Account profile	Microsoft in education	Azure	Microsoft Visual Studio	Careers
Surface Laptop 2	Download Center	Office for students	AppSource	Windows Dev Center	About Microsoft
Surface Go	Microsoft Store support	Office 365 for schools	Automotive	Developer Network	Company news
Xbox One X	Returns	Deals for students & parents	Government	TechNet	Privacy at Microsoft
Xbox One S	Order tracking	Microsoft Azure in education	Healthcare	Microsoft developer program	Investors
VR & mixed reality	Store locations		Manufacturing	Channel 9	Diversity and inclusion
Windows 10 apps	Buy online, pick up in store		Financial services	Office Dev Center	Accessibility
Office apps			Retail	Microsoft Garage	Security

Site Feedback

 English (United States)

[Sitemap](#) [Contact Microsoft](#) [Privacy & cookies](#) [Terms of use](#) [Trademarks](#) [Safety & eco](#) [About our ads](#)

© Microsoft 2019